Вычислительный практикум (9-ый семестр, 2021)

Дифференциальные уравнения параболического типа

Руководитель: А. Г. Доронина *Выполнил: П. Л. Соболев*

Задачи

- Создать алгоритм решения дифференциального уравнения параболического типа при заданных видах дифференциального оператора и граничных условий и протестировать его на выбранном решении.

Теория

Постановка задачи: найти решение $u(x,t)$ проблемы

$$\frac{\partial u}{\partial t} = Lu + f(x,t),\ 0 < x < 1,\ 0 < t \leqslant T, \tag{1}$$

$$u(x,0) = \varphi(x),\ 0 \leqslant x \leqslant 1, \tag{2}$$

$$\left.\left(\alpha_1(t)u - \alpha_2(t)\frac{\partial u}{\partial x}\right)\right|_{x=0} = \alpha(t),\ 0 \leqslant t \leqslant 0.1, \tag{3}$$

$$\left.\left(\beta_1(t)u + \beta_2(t)\frac{\partial u}{\partial x}\right)\right|_{x=0} = \beta(t),\ 0 \leqslant t \leqslant 0.1, \tag{4}$$

где

$$Lu = \left[\begin{array}{l} a(x,t)\dfrac{\partial^2 u}{\partial x^2} + b(x,t)\dfrac{\partial u}{\partial x} + c(x,t)u; \\ \dfrac{\partial}{\partial x}\left(p(x)\dfrac{\partial u}{\partial x}\right) + b(x,t)\dfrac{\partial u}{\partial x} + c(x,t)u. \end{array}\right. \tag{5}$$

Разностный аналог дифференциального оператора (5) на равномерной сетке $\big\{(x_i = hi, t_k = \tau k),\ h = 1/N,\ \tau = 1/M,\ i = \overline{0,N},\ k = \overline{0,M}\big\}$:

$$L_h u_i^k = \left[\begin{array}{l} a(x_i,t_k)\frac{u_{i+1}^k - 2u_i^k + u_{i-1}^k}{h^2} + b(x_i,t_k)\frac{u_{i+1}^k - u_{i-1}^k}{2h} + c(x_i,t_k)u_i^k; \\ p_{i+\frac{1}{2}}\frac{u_{i+1}^k - u_i^k}{h^2} - p_{i-\frac{1}{2}}\frac{u_i^k - u_{i-1}^k}{h^2} + b(x_i,t_k)\frac{u_{i+1}^k - u_{i-1}^k}{2h} + c(x_i,t_k)u_i^k. \end{array}\right. \tag{6}$$

Рассмотрим однопараметрическое семейство разностных схем:

$$\begin{gathered} \frac{u_i^k - u_i^{k-1}}{\tau} = L_h(\sigma u_i^k + (1-\sigma)u_i^{k-1}) + f(x, \overline{t_k}), \\ \sigma \in [0,1],\ \overline{t_k} = t_k + (1-\sigma)\tau,\ i = \overline{1, N-1},\ k = \overline{1,M}. \end{gathered} \tag{7}$$

Из начального условия (2) имеем

$$u_i^0 = \varphi(x_i),\ i = \overline{0,N}. \tag{8}$$

Если $\sigma = 0$, то разностная схема (7) является явной. Аппроксимируя производные в граничных условиях (3)–(4) с порядком $O(h^2)$, получаем следующие выражения для вычисления значений решения при $k = \overline{1,M}$:

$$u_i = u_i^{k-1} + \tau(L_h u_i^{k-1} + f(x_i, t_{k-1})),\ i = \overline{1, N-1}, \tag{9}$$

$$\alpha_1(t_k)u_0^k - \alpha_2(t_k)\frac{-3u_0^k + 4u_1^k - u_2^k}{2h} = \alpha(t_k), \tag{10}$$

$$\beta_1(t_k)u_N^k + \beta_2(t_k)\frac{3u_N^k - 4u_{N-1}^k + u_{N-2}^k}{2h} = \beta(t_k). \tag{11}$$

Условие устойчивости явной разностной схемы:

$$A\nu \leqslant \frac{1}{2}, \tag{12}$$

где $\nu = \tau/h^2$, а

$$A = \left[\begin{array}{l}\max\{a(x,t) \mid 0 \leqslant x \leqslant 1,\ 0 \leqslant t \leqslant T\}; \\ \max\{p(x) \mid 0 \leqslant x \leqslant 1\}.\end{array}\right. \tag{13}$$

Если $\sigma \neq 0$, то разностная схема (7) является неявной и может быть представлена в виде

$$\sigma L_h u_i^k - \frac{1}{\tau}u_i^k = G_i^k, \tag{14}$$

где

$$G_i^k = -\frac{1}{\tau}u_i^{k-1} - (1-\sigma)L_h u_i^{k-1} - f(x_i, \overline{t_k}),\ i = \overline{1, N-1},\ k = \overline{1, M}. \tag{15}$$

Аппроксимируя производные в граничных условиях (3)–(4) с порядком $O(h)$, получаем

$$\alpha_1(t_k)u_0^k - \alpha_2(t_k)\frac{u_1^k - u_0^k}{h} = \alpha(t_k), \tag{16}$$

$$\beta_1(t_k)u_N^k + \beta_2(t_k)\frac{u_N^k - u_{N-1}^k}{h} = \beta(t_k). \tag{17}$$

Приведя граничные условия (16)–(17) к виду

$$\begin{gathered} -B_0 u_0^k + C_0 u_1^k = G_0^k, \\ A_N u_{N-1}^k - B_N u_N^k = G_N^k \end{gathered} \tag{18}$$

и объединив их в систему с выражениями (14)–(15), получаем следующую систему линейных уравнений для вычисления значений решения при $k = \overline{1, M}$:

$$\begin{aligned} - B_0 u_0^k \ &+ C_0 u_1^k &&= G_0^k, \\ A_i u_{i-1}^k - B_i u_i^k \ &+ C_i u_{i+1}^k &&= G_i^k,\ i = \overline{1, N-1}, \\ A_N u_{N-1}^k - B_N u_N^k & &&= G_N^k. \end{aligned} \tag{19}$$

Реализация

Алгоритм реализован на языке программирования Julia в виде скрипта и расположен в GitLab репозитории Computational Workshop S09-2021 в папке A2. Для воспроизведения результатов следуй инструкциям в файле `README.md`.

Проблема:

$$\begin{gathered}\frac{\partial u}{\partial t}=\frac{\partial}{\partial x}\left((x+3)\frac{\partial u}{\partial x}\right)-xu+f(x,t),\\ u(x,0)=\varphi(x),\ 0\leqslant x\leqslant 1,\\ \left.\frac{\partial u}{\partial x}\right|_{x=0}=\alpha(t),\ u(1,t)=\beta(t),\ 0\leqslant t\leqslant 0.1.\end{gathered}\tag{20}$$

Для проверки алгоритма возьмем решение $u(x,t)=x+t$.

Листинг 1: Определение функций дифф. уравнения и проблемы

```
# Define the differential equation's functions
p(x) = x + 3
b(_, _) = 0
c(x, _) = -x
α₁(t) = 0
α₂(t) = -1
β₁(t) = 1
β₂(t) = 0

# Define the problem's functions
u(x, t) = x + 3t
f(x, t) = x^2 + 3 * x * t + 2
φ(x) = x
α(t) = 1
β(t) = 1 + 3t
```

Листинг 2: Поиск сетки, дающей стабильное решение для явной схемы

```
"Check if a solution will be stable on the specified grid"
function is_stable(N, M)::Bool
    A = maximum(p, 0:0.0001:1)
    h = 1 / N
    τ = T / M
    ν = τ / h^2
    return A * ν ≤ 0.5
end

"Using provided N, find M, which gives a stable solution"
function find_stable(N)::Int
    M = 5
    while !is_stable(N, M)
        M *= 2
    end
    return M
end
```

Листинг 3: Вычисление дифференциального оператора L

```
"Calculate the value of the differential operator L"
function L(um, x, t, h, i, k)::Float64
    return p(x[i] + h/2) * (um[k,i+1] - um[k,i]) / h^2 -
           p(x[i] - h/2) * (um[k,i] - um[k,i-1]) / h^2 +
           b(x[i], t[k]) * (um[k,i+1] - um[k,i-1]) / (2h) +
           c(x[i], t[k]) * um[k, i]
end
```

Листинг 4: Вычисление решения по явной схеме

```
# <...>
# Prepare a matrix for the solution
# (number of columns is the number of nodes for x's)
um = Matrix{Float64}(undef, M + 1, N + 1)
# Compute the first layer
um[1, :] .= φ.(x)
# Use the explicit scheme for other layers
for k = 2:M+1
    # Compute for i in 2:N
    for i = 2:N
        um[k, i] = um[k-1,i] + τ * f(x[i],t[k-1]) +
        τ * L(um, x, t, h, i, k - 1)
    end
    # Compute the rest
    um[k, 1] = (α(t[k]) + α₂(t[k]) *
               (4 * um[k, 2] - um[k, 3]) / (2h)) /
               (α₁(t[k]) + 3 * α₂(t[k]) / (2h))
    um[k, N+1] = (β(t[k]) + β₂(t[k]) *
                 (4 * um[k, N] - um[k, N-1]) / (2h)) /
                 (β₁(t[k]) + 3 * β₂(t[k]) / (2h))
end
# <...>
```

Листинг 5: Вычисление решения по неявной схеме

```
# <...>
# Prepare a matrix for the solution
# (number of columns is the number of nodes for x's)
um = Matrix{Float64}(undef, M + 1, N + 1)
# Compute the first layer
um[1, :] .= φ.(x)
# Use the implicit scheme for other layers
for k = 2:M+1
    # Compute the linear system's matrix
    tm = Tridiagonal(
        # A's
        [
            [
                σ * (p(x[i] - h / 2) / h^2 -
                b(x[i], t[k]) / (2h))
                for i = 2:N
            ]; -β₂(t[k]) / h
        ],
        # B's
        [
            α₁(t[k]) + α₂(t[k]) / h; -[
                1 / τ + σ * ((p(x[i] + h / 2) +
                p(x[i] - h / 2)) / h^2 -
                c(x[i], t[i]))
                for i = 2:N
            ]; β₁(t[k]) + β₂(t[k]) / h
        ],
        # C's
        [
            -α₂(t[k]) / h; [
              σ * (p(x[i] + h / 2) / h^2 +
              b(x[i], t[k]) / (2h))
              for i = 2:N
            ]
        ],
    )
    # Compute the linear system's right-hand side vector
    g = [
        α(t[k]); [
          -um[k-1, i] / τ - (1 - σ) *
          L(um, x, t, h, i, k - 1) -
          f(x[i], t[k] - (1 - σ) * τ)
          for i = 2:N
        ]; β(t[k])
    ]
    # Compute the solution of the linear system
    um[k, :] = tm \ g
end
```

Таблица 1: Решение дифференциального уравнения

$t \backslash x$	0	0.2	0.4	0.6	0.8	1
0	0.0	0.2	0.4	0.6	0.8	1.0
0.02	0.06	0.26	0.46	0.66	0.86	1.06
0.04	0.12	0.32	0.52	0.72	0.92	1.12
0.06	0.18	0.38	0.58	0.78	0.98	1.18
0.08	0.24	0.44	0.64	0.84	1.04	1.24
0.1	0.3	0.5	0.7	0.9	1.1	1.3

Таблица 2: Сравнение результата с точным решением на явной схеме

h	τ	Δu
0.2	0.05	$2.220446049250313 \cdot 10^{-16}$
0.1	0.0125	$3.608224830031759 \cdot 10^{-16}$
0.05	0.003125	$5.551115123125783 \cdot 10^{-16}$

Таблица 3: Сравнение результата с точным решением на неявной схеме

h	τ	Δu ($\sigma = 0.5$)	Δu ($\sigma = 1.0$)
0.2	0.001	$5.639932965095795 \cdot 10^{-14}$	$2.403632848313464 \cdot 10^{-14}$
0.1	0.001	$1.343369859796439 \cdot 10^{-14}$	$3.758104938356155 \cdot 10^{-14}$
0.05	0.001	$6.361577931102147 \cdot 10^{-14}$	$2.675637489346627 \cdot 10^{-14}$